%%   The contents of this file are subject to the Mozilla Public License
%%   Version 1.1 (the "License"); you may not use this file except in
%%   compliance with the License. You may obtain a copy of the License at
%%   http://www.mozilla.org/MPL/
%%
%%   Software distributed under the License is distributed on an "AS IS"
%%   basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
%%   License for the specific language governing rights and limitations
%%   under the License.
%%
%%   The Original Code is RabbitMQ Management Plugin.
%%
%%   The Initial Developer of the Original Code is GoPivotal, Inc.
%%   Copyright (c) 2010-2014 GoPivotal, Inc.  All rights reserved.

-module(rabbit_mgmt_wm_consumers).

-export([init/1, to_json/2, content_types_provided/2, resource_exists/2,
         is_authorized/2]).

-import(rabbit_misc, [pget/2]).

-include("rabbit_mgmt.hrl").
-include("webmachine.hrl").
-include("rabbit.hrl").

%%--------------------------------------------------------------------

init(_Config) -> {ok, #context{}}.

content_types_provided(ReqData, Context) ->
   {[{"application/json", to_json}], ReqData, Context}.

resource_exists(ReqData, Context) ->
    {case rabbit_mgmt_util:vhost(ReqData) of
         vhost_not_found -> false;
         _               -> true
     end, ReqData, Context}.

to_json(ReqData, Context = #context{user = User}) ->
    Consumers = case rabbit_mgmt_util:vhost(ReqData) of
                    none  -> rabbit_mgmt_db:get_all_consumers();
                    VHost -> rabbit_mgmt_db:get_all_consumers(VHost)
                end,
    rabbit_mgmt_util:reply_list(
      filter_user(Consumers, User), ReqData, Context).

is_authorized(ReqData, Context) ->
    rabbit_mgmt_util:is_authorized(ReqData, Context).

filter_user(List, #user{username = Username, tags = Tags}) ->
    case rabbit_mgmt_util:is_monitor(Tags) of
        true  -> List;
        false -> [I || I <- List,
                       pget(user, pget(channel_details, I)) == Username]
    end.
